
Solutions to Assignment 1
VHDL Synthesis

The entity should look something like:

-- ELEC 464 Assignment 1
-- model of 74’151 8 to 1 multiplexer with
-- complementary outputs and enable
-- Ed Casas, September 10 1996

entity EC151 is
port ( a, b, c, s : in bit ;

d : in bit_vector (7 downto 0) ;
y, w : out bit ) ;

end EC151 ;

A straighforward architecture written with
if/then/else statements might look like:

architecture rtl of EC151 is
begin

process(a,b,c,s,d)
begin

if s = ’0’ then
y <= ’0’ ;
w <= ’1’ ;

else
if c=’0’ and b=’0’ and a=’0’ then

y <= d(0) ;
w <= not d(0) ;

elsif c=’0’ and b=’0’ and a=’1’ then
y <= d(1) ;
w <= not d(1) ;

elsif c=’0’ and b=’1’ and a=’0’ then
y <= d(2) ;
w <= not d(2) ;

elsif c=’0’ and b=’1’ and a=’1’ then
y <= d(3) ;
w <= not d(3) ;

elsif c=’1’ and b=’0’ and a=’0’ then
y <= d(4) ;
w <= not d(4) ;

elsif c=’1’ and b=’0’ and a=’1’ then
y <= d(5) ;
w <= not d(5) ;

elsif c=’1’ and b=’1’ and a=’0’ then
y <= d(6) ;
w <= not d(6) ;

else
y <= d(7) ;
w <= not d(7) ;

end if ;
end if ;

end process ;

end rtl ;

The result of synthesizing this description is:

An alternative architecture can be written by de-
riving the sum-of-products expressions for y and w
from a truth table containing all the possible combi-
nations of inputs. The description can be simplified
by using a separate process to compute w and y from
an intermediate internal signal:

architecture rtl of EC151 is
-- internal (temporary) signal:
signal tmp : bit ;

begin

process(a,b,c,d)
begin

tmp <=
( not c and not b and not a and d(0) ) or
( not c and not b and a and d(1) ) or
( not c and b and not a and d(2) ) or
( not c and b and a and d(3) ) or
( c and not b and not a and d(4) ) or
( c and not b and a and d(5) ) or
( c and b and not a and d(6) ) or
( c and b and a and d(7) ) ;

end process ;

process(tmp,s)
begin

y <= s and tmp ;
w <= not ( s and tmp ) ;

end process ;

end rtl ;

Note that the parentheses are required because the
and and or logical operators have the same prece-
dence.

The result of synthesizing this description is:

A third approach is to combine a, b, and c into
an internal 3-element bit_vector and use it in a
case statement:

architecture rtl of EC151 is
-- internal signals:
signal tmp : bit ;
signal sel : bit_vector (2 downto 0) ;

begin

process(a,b,c)
begin

sel(2) <= c ;
sel(1) <= b ;
sel(0) <= a ;

end process ;

process(sel,d)
begin

case sel is
when "000" => tmp <= d(0) ;
when "001" => tmp <= d(1) ;
when "010" => tmp <= d(2) ;
when "011" => tmp <= d(3) ;
when "100" => tmp <= d(4) ;
when "101" => tmp <= d(5) ;
when "110" => tmp <= d(6) ;
when "111" => tmp <= d(7) ;
end case ;

end process ;

process(tmp,s)
begin

y <= s and tmp ;
w <= not ( s and tmp ) ;

end process ;

end rtl ;

This architecture synthesizes to the same circuit as
the previous architecture.

The final approach uses some features of VHDL
that we have not covered in the lectures. Three bit
values are combined into a bit_vector and con-
verted to an integer using a vendor-dependent (non-
portable) conversion function (bvtoi). This integer
expression is used to index d. We also use concur-
rent signal assignments.

library synopsys ;
use synopsys.bv_arithmetic.all ;

architecture rtl of EC151 is
signal tmp : bit ;

begin

tmp <= s and d(bvtoi((c,b,a))) ;
y <= tmp ;
w <= not tmp ;

end rtl ;

This architecture also synthesizes to the same cir-
cuit as the previous architecture. The main draw-
back of this solution is that it will not work with
other companies’ VHDL software. Similar but more
portable solutions are available using functions from
IEEE VHDL libraries.

